Lagrange Points

1 Discussion

> No diagrams will be found in this work.
>
> Joseph-Louis Lagrange, in the preface of his *Mécanique analytique* published 1788

While Joseph-Louis Lagrange (1736–1813) was proud to dispose of the geometrical language which Newton had used to write his *Principia*, we will discuss Lagrange's mechanics using diagrams. Many of Lagrange's efforts were directed at the *Three Body Problem*. While the motion of two point masses interacting through their mutual gravitational attraction is no more complicated than the motion of a single point-mass in a fixed $\frac{1}{r^2}$ force-field, the motion of three bodies interacting through gravity is a singularly difficult problem. Even today solutions take the form of approximations, rather than analytical results. Newton in working on the prototypical three-body problem: the motion of Sun, Earth, and Moon, said that it "made his head ache". (We will mostly phrase our work in terms of three bodies: Earth, Moon, and satellite.)

We begin by considering the motion of the two most massive bodies (for us the Earth and Moon) and then move on to consider the motion of an object so light that it does not affect its partners. (Notice that this is a violation of Newton's Third Law!) We will allow Newton's Third to play a role for our two big bodies. Thus Moon pulls on the Earth exactly as strongly as the the Earth pulls on the Moon. . . the Earth is not an unmoved mover; it is also in orbit. The fixed point in the interaction between Earth an Moon is the center of mass.





We assume circular motion with (shared) angular speed ω, so:

$$M_1 r_1 = M_2 r_2$$

$$M_1\omega^2 r_1 = \frac{GM_1M_2}{(r_1+r_2)^2} \qquad M_2\omega^2 r_2 = \frac{GM_1M_2}{(r_1+r_2)^2}$$

$$\omega^2 r_1 = \frac{GM_2}{(r_1+r_2)^2} \qquad \omega^2 r_2 = \frac{GM_1}{(r_1+r_2)^2}$$

$$\omega^2(r_1+r_2) = \frac{G(M_1+M_2)}{(r_1+r_2)^2}$$

$$\omega^2 = \frac{G(M_1+M_2)}{(r_1+r_2)^3}$$

We now consider the view from a frame of reference rotating about the CM at exactly the pace needed to keep up with the Moon. In this frame, both Earth and Moon are at rest, but there are addition pseudo forces: centrifugal and Coriolis[1]. A satellite moving in our new frame will follow:

$$m\mathbf{a} = -\frac{GM_1 m}{|\mathbf{r}-\mathbf{r}_1|^2}\mathbf{u_1} - \frac{GM_2 m}{|\mathbf{r}-\mathbf{r}_2|^2}\mathbf{u_2} + m\omega^2\mathbf{r} + 2m\omega\mathbf{v}\times\hat{\mathbf{z}}$$

Notice that the Coriolis force is just like the $\mathbf{v}\times\mathbf{B}$ magnetic force, the outward centrifugal force is like a sign-reversed spring force, and all forces are proportional to m. We now switch to units scaled to the problem. If T is the period of the Moon's orbit (i.e., $\omega = 2\pi/T$), then our dimensionless version of time, t', is defined by:

$$t = t'T$$

Similarly we define a dimensionless distance:

$$\mathbf{r} = \mathbf{r}'(r_1+r_2) = \mathbf{r}'d$$

If we switch to these variables our satellite's acceleration is given by:

$$\begin{aligned}
\frac{d}{T^2}\,\mathbf{a}' &= \frac{G(M_1+M_2)}{d^2}\left(-\frac{M_1/(M_1+M_2)}{|\mathbf{r}'-\mathbf{r}_1'|^2}\mathbf{u_1} - \frac{M_2/(M_1+M_2)}{|\mathbf{r}'-\mathbf{r}_2'|^2}\mathbf{u_2}\right) + \omega^2 d\left(\mathbf{r}' + 2\frac{1}{T\omega}\mathbf{v}'\times\hat{\mathbf{z}}\right) \\
&= \omega^2 d\left(-\frac{1-\epsilon}{|\mathbf{r}'-\mathbf{r}_1'|^2}\mathbf{u_1} - \frac{\epsilon}{|\mathbf{r}'-\mathbf{r}_2'|^2}\mathbf{u_2} + \mathbf{r}' + \frac{1}{\pi}\mathbf{v}'\times\hat{\mathbf{z}}\right) \\
\mathbf{a}' &= (2\pi)^2\left(-\frac{1-\epsilon}{|\mathbf{r}'-\mathbf{r}_1'|^2}\mathbf{u_1} - \frac{\epsilon}{|\mathbf{r}'-\mathbf{r}_2'|^2}\mathbf{u_2} + \mathbf{r}' + \frac{1}{\pi}\mathbf{v}'\times\hat{\mathbf{z}}\right)
\end{aligned}$$

where $\epsilon = M_2/(M_1+M_2)$. (For the Earth-Moon system: $\epsilon = .0121$.) Except for the velocity-dependent Coriolis force we can make a potential to describe force ($-\nabla\phi = \mathbf{a}'$):

$$\phi(\mathbf{r}') = (2\pi)^2\left(-\frac{1-\epsilon}{|\mathbf{r}'-\mathbf{r}_1'|} - \frac{\epsilon}{|\mathbf{r}'-\mathbf{r}_2'|} - \frac{1}{2}\,r'^2\right)$$

I'm tired of putting primes on everything; in what follows $\mathbf{r}$ stands for $\mathbf{r}'$, etc.

```
phi[x_,y_,e_]=(2 Pi)^2 ( -(1-e)/Sqrt[(x+e)^2+y^2] -e/Sqrt[(x-(1-e))^2+y^2] -
     (1/2)(x^2+y^2)    )
fx[x_,y_,e_]=-D[phi[x,y,e],x]
fy[x_,y_,e_]=-D[phi[x,y,e],y]

ContourPlot[phi[x,y,.1],{x,-1.5,1.5},{y,-1.2,1.2},
Contours->10,PlotPoints->50,PlotRange->{-80,-50},AspectRatio->Automatic]
```

The contour plot shows a view similar to that looking down the cone of a double-sourced volcano. Well away from the origin, the centrifugal potential pushes everything out—that way is down the sides of the volcano cone. There are two "holes" which might source lava in a volcanic eruption—that way is down the gravitational well either to M_1 or M_2. Around the edge of the cone is the crater lip. The lowest exit out is along the positive x axis (the connection between M_1 and M_2 is a bit lower), the exit along the negative x axis is the highest valley between the twin peaks at $\pm 60°$ from M_1. This shows this volcano:

[1] France: Gustav-Gaspard Coriolis, 1835




```
Plot3D[phi[x,y,.1],{x,-1.5,1.5},{y,-1.5,1.5},
PlotPoints->50,PlotRange->{-80,-50},ViewPoint->{2,-2,.6}]
```

Now to have a "stationary" satellite (in this rotating frame!) we must be at a point where $\mathbf{F} = \mathbf{0}$. Maximums, minimums, and "saddle points" provide $\nabla\phi = \mathbf{0}$. The saddle points don't look like stable equilibrium points; the peaks look even less promising. . . but don't forget that "magnetic field". You should know that physicists (and Scotty on Star Trek) use magnetic fields to confine particles. . . Maybe it will stabilize our satellite!

But first let's find the exact location of the peaks:

```
xp=-e+Cos[Pi/3]
yp=Sin[Pi/3]
Simplify[fx[xp,yp,e]]
Simplify[fy[xp,yp,e]]
```

The point indeed has $\mathbf{F} = \mathbf{0}$. From the origin, the point is back ϵ (i.e., on top of M_1), and then exactly 1 unit 60° from the x-axis. This proves that M_1, M_2, and our peak make an equilateral triangle. You might have hoped *Mathematica* could directly find the roots with a command like:

```
 Solve[{fx[x,y,e]==0,fy[x,y,e]==0},{x,y}]
```

but the problem is too complex for *Mathematica* to find the roots without some help. In an appendix we will demonstrate the peaks are stable equilibrium positions for $\epsilon < 0.0385$. Instead let's view satellite motion

```
solution=NDSolve[{x''[t]==fx[x[t],y[t],.01]+(4 Pi) y'[t], y''[t]==fy[x[t],y[t],.01]-(4 Pi) x'[t],
    x[0]==-.01+Cos[Pi/3]+.01, y[0]==Sin[Pi/3],
    x'[0]==0,          y'[0]==0}, {x,y},{t,0,10}]

Out[11]= {{x -> InterpolatingFunction[{{0., 10.}}, <>],
 >     y -> InterpolatingFunction[{{0., 10.}}, <>]}}

ParametricPlot[Evaluate[{x[t],y[t]} /. solution],{t,0,10}]
ContourPlot[phi[x,y,.01],{x,0,1.3},{y,0,1.3},ContourShading -> False,
Contours->20,PlotPoints->50,PlotRange->{-65,-59},AspectRatio->Automatic]
Show[%%,%,AspectRatio->Automatic]

solution=NDSolve[{x''[t]==fx[x[t],y[t],.04]+(4 Pi) y'[t], y''[t]==fy[x[t],y[t],.04]-(4 Pi) x'[t],
    x[0]==-.04+Cos[Pi/3]+.001, y[0]==Sin[Pi/3],
```

```
    x'[0]==0,          y'[0]==0}, {x,y},{t,0,10}]

Out[17]= {{x -> InterpolatingFunction[{{0., 10.}}, <>],
>     y -> InterpolatingFunction[{{0., 10.}}, <>]}}

ParametricPlot[Evaluate[{x[t],y[t]} /. solution],{t,0,10}]
ContourPlot[phi[x,y,.01],{x,0,1.3},{y,0,1.3},ContourShading -> False,
Contours->20,PlotPoints->50,PlotRange->{-65,-59},AspectRatio->Automatic]
Show[%%,%,AspectRatio->Automatic]
```





Thus with $\epsilon = .04$ and initial condition very close to the point where $\mathbf{F} = \mathbf{0}$ (above right) the motion seems to be diverging, whereas with $\epsilon = .01$ and initial condition further from the $\mathbf{F} = \mathbf{0}$ location (above left) the motion seems confined.

2 Lab

Remember to turn in a printout showing each step as *Mathematica* solves the problem, in addition to any requested plots. For a review of basic orbit parameters see:

`http://www.physics.csbsju.edu/orbit/orbit.2d.html`

Option A:

Moon Shot: starting from $\mathbf{r} = (.1, 0)$, find the velocity needed to project the spacecraft around the Moon and back in the general direction of the Earth. Hint: a good starting point is to find the escape velocity from that location. Turn in a plot showing your trajectory.

Option B:

Moon Perturbations: Without the Moon, an Earth satellite orbits the Earth in a fixed ellipse. To what extent does the Moon's pull change this result? Starting from $\mathbf{r} = (.4, 0)$, view an Earth orbit in the inertial frame. You must back-rotate to reach the inertial frame. At the same time it is useful to view the location relative to the Earth rather than the CM:

```
rot[t_]={{Cos[2 Pi t], -Sin[2 Pi t]},{Sin[2 Pi t],Cos[2 Pi t]}}
v0=3; theta=Pi/2
solution=NDSolve[{x''[t]==fx[x[t],y[t],.0121]+(4 Pi) y'[t],
    y''[t]==fy[x[t],y[t],.0121]-(4 Pi) x'[t],
    x[0]==.4, y[0]==0, x'[0]==v0 Cos[theta], y'[0]==v0 Sin[theta]},
```

```
    {x,y},{t,0,12},MaxSteps -> 50000]

ParametricPlot[Evaluate[rot[t].{x[t]+.0121,y[t]} /. solution],{t,0,.1},
AspectRatio->Automatic]
ParametricPlot[Evaluate[rot[t].{x[t]+.0121,y[t]} /. solution],{t,11.9,12},
AspectRatio->Automatic,PlotStyle->{RGBColor[1,0,0]}]
Show[%,%%]
```

These plots show the original orbit and the orbit one year later. Tape hardcopies into your notebook. From your plots and `solution` determine: the semi-major axis of the orbit, the period of the orbit, and the *precession* (changing orientation of the orbit: unit= orbit rotation angle per year). Using Kepler's Third Law calculate the orbit period.

Option C:

Consider the precessing orbit, calculated for a year, described in Option B above. At any instant the orientation and eccentricity of the orbit can be calculated from the Lenz vector[2]:

$$\mathbf{A} = \frac{(\mathbf{r} \times \mathbf{v}) \times \mathbf{v}}{(2\pi)^2} + \frac{\mathbf{r}}{|\mathbf{r}|}$$

In this formula $\mathbf{r}$ and $\mathbf{v}$ are measured in an inertial frame with the Earth at the origin, so we must displace our CM-centered coordinate $\mathbf{r}$ to center on the Earth, add the velocity from rotation to the velocity measured in the rotating frame, and "unrotate" the vectors.

```
vx[t_]=x'[t]-y[t]2 Pi /. First[solution]
vy[t_]=y'[t]+x[t]2 Pi /. First[solution]
xe[t_]=x[t]+.0121 /. First[solution]

lenz=rot[t].({vy[t](y[t]vx[t]-xe[t]vy[t]),vx[t](xe[t]vy[t]-y[t]vx[t])}/(2 Pi)^2 +
{xe[t],y[t]}/Sqrt[xe[t]^2+y[t]^2]) /. First[solution]
```

The magnitude of $\mathbf{A}$ is the eccentricity; the direction of $\mathbf{A}$ is apogee. By measuring on a hardcopy plot of an orbit, determine the eccentricity. `ParametricPlot lenz` (hardcopies please!) during the first and last month. Note the varying eccentricity and direction-to-apogee during <u>an</u> orbit. Note the shift in average $\mathbf{A}$ over a year. Use this shift to calculate the precession rate (see Option B). Calculate the magnitude of $\mathbf{A}$ and compare to the eccentricity found from an orbit plot.

Option D:

Numerical Solution Errors: In contrast to *Mathematica*'s symbol manipulation, *Mathematica*'s numerical calculations have (unavoidable) error. We know that energy is conserved, but you will find that in the numerical solution energy is not conserved. Work out the "Option B" orbit and plot out energy as a function of time:

```
phi[x[t],y[t],.0121]+(1/2)(x'[t]^2+y'[t]^2) /. First[solution]
Plot[%,{t,0,1}]
Plot[%%,{t,11,12}]
```

Convert the average percentage change in energy per year to a percentage change in semi-major axis per orbit. Could you see this error in the orbit plot? The "numerical" friction will eventually bring the satellite down. Assuming the eccentricity and the percentage decrease in a remain constant, find when this orbit will intersect the Earth (at perigee of course).

[2]Lenz refers to W. Lenz who used this vector in his 1924 paper on the H-atom. However, Laplace's 1799 work *Traité de mécanique celeste* contains a much earlier use of this "vector"

Option E:

Tides: A fluid Earth would conform to an isopotential surface. In general the effect is small so we must "un-wrap" the Earth before we plot (N.B. radius of Earth=.0166; show this!):

```
ContourPlot[phi[(r+.0166)Cos[u]-.0121,(r+.0166)Sin[u],.0121],{u,-Pi,Pi},{r,0,.00000001}]
```

The above plots the isopotential lines for a range of about 10 feet above a sphere centered on the Earth. Note that the isopotential stretches out about a foot at $u = \pm\pi, 0\ldots$ A fluid Earth would be extended towards the Moon ($u = 0$) and in the direction opposite the Moon ($u = \pm\pi$). Explain why this means we should have about two high tides per day. Show that the size of the tides is approximately as I stated above.

Option F:

Binary Stars & Mass Transfer: Consider a binary star system where $M_1 = 2M_2$, i.e., $\epsilon = \frac{1}{3}$. It turns out that, if these stars were born at the same time, M_1 will expand to become a red giant first. If the stars are separated by less than the distance M_1 seeks to expand into, some of the gas that makes up M_1 will be deposited onto M_2. As stated above, fluid (e.g., gaseous) objects will try to conform to isopotentials. Thus as M_1 starts to expand, the low potential point between the stars will serve as the lip pouring material onto M_2 (i.e., surplus material for M_1 will start from the Lagrange point L_1 on a trip towards M_2. Find the L_1 location. Perform a numerical solution to the differential equation starting material a bit beyond L_1 with zero velocity. Find motion of the material over a month. `ParametricPlot` the trajectory on top of the isopotentials contours (as on p. 4). What M_2 radius would result in the material just barely hitting "on the first try"? (Think about plotting distance to M_2 vs. time.)

3 Appendix

We seek here to demonstrate the stability of the L_4 Lagrange point. We begin by Taylor expanding the potential in the vicinity of L_4. Clearly near L_4 the potential must have the form:

$$\phi = \phi_0 + \frac{1}{2}Ax^2 + Bxy + \frac{1}{2}Cy^2$$

ϕ_0 has no effect on the forces, linear terms in the Taylor expansion are zero since by definition $\nabla\phi = \mathbf{0}$ there, and we neglect the higher order terms which will be quite small a few km (i.e., $\Delta x, \Delta y \sim 10^{-6}$) from L_4. It turns out that we will have to work some to extract the homogeneous quadratic piece of the Taylor polynomial. If you remember your high school analytical geometry you should remember that by a suitable rotation of axes, the above form can be transformed to:

$$\phi = \phi_0 + \frac{1}{2}A'x'^2 + \frac{1}{2}C'y'^2$$

```
Series[phi[x-e+Cos[Pi/3],y+Sin[Pi/3],e],{x,0,2},{y,0,2}]
```

```
Normal[%] /.  {x->q x, y->q y}
```

…so the terms we want will be q^2

```
Coefficient[%, q^2]
```

…OK we've got our quadratic form; now rotate axes

The easiest way to rotate our axes is to express the quadratic form in matrix form:

$$\frac{1}{2}Ax^2 + Bxy + \frac{1}{2}Cy^2 = \frac{1}{2}\begin{pmatrix} x & y \end{pmatrix}\begin{pmatrix} A & B \\ B & C \end{pmatrix}\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} x & y \end{pmatrix} M \begin{pmatrix} x \\ y \end{pmatrix}$$

and then diagonalize M (i.e., find it's eigenvalues). The orthogonal matrix that diagonalizes M also rotates the coordinate system.

```
m={{Coefficient[%, x^2],Coefficient[%, x y]/2},{ Coefficient[%, x y]/2, Coefficient[%, y^2]}}
Eigensystem[%]
mdiag=DiagonalMatrix[First[%]]
phiApprox[x_,y_,e_]={y,x}.mdiag.{y,x}
fxApprox[x_,y_,e_]=-D[phiApprox[x,y,e],x]
fyApprox[x_,y_,e_]=-D[phiApprox[x,y,e],y]

DSolve[{x''[t]==fxApprox[x[t],y[t],e]+(4 Pi) y'[t], y''[t]==fyApprox[x[t],y[t],e]-(4 Pi) x'[t]},
   {x,y},t]
```

Mathematica's solution goes on for some time, but you will notice the following terms:

```
                   2                        2    2
  Sqrt[2] Sqrt[-Pi  + Sqrt[1 - 27 e + 27 e ] Pi ] #1
E

                   2                        2    2
 I Sqrt[2] Sqrt[Pi  + Sqrt[1 - 27 e + 27 e ] Pi ] #1
E
```

Clearly the nature of the solution changes if

$$1 - 27\epsilon + 27\epsilon^2 < 0$$

as it will be if $0.0385 \approx (9 - \sqrt{69})/18 < \epsilon < (9 + \sqrt{69})/18 \approx .9615$

Note there is nothing particularly difficult in solving this approximate differential equation:

$$\ddot{\mathbf{r}} = \begin{pmatrix} A' & 0 \\ 0 & C' \end{pmatrix} \cdot \mathbf{r} + 4\pi \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix} \cdot \dot{\mathbf{r}}$$

It can be solved by assuming a solution of the form: $\mathbf{r} = \mathbf{r}_0\, e^{\gamma t}$, and then solving the "eigenproblem":

$$\begin{pmatrix} A' - \gamma^2 & 4\pi\gamma \\ -4\pi\gamma & C' - \gamma^2 \end{pmatrix} \cdot \mathbf{r}_0 = 0$$